AD-A117 533 ROYAL AIRCRAFT ESTABLISHMENT FARNBOROUGH (ENGLAND) F/G 20/4
WAVE DRAG FROM OBLIQUE CROSS-SECTIONAL AREAS.(U)
JAN 82 J PIKE
UNCLASSIFIED RAE-TR-81155 DRIC-BR-83415 NL




1.0
1.1
1.25
1.4
1.6
1.8
2.0
2.2
2.5
2.8
3.2
3.6
4.0
4.5
5.0
5.6
6.3
MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

ROYAL AIRCRAFT ESTABLISHMENT

*

Technical Report 81155

January 1982

WAVE DRAG FROM OBLIQUE CROSS-SECTIONAL AREAS

by

J. Pike

*

UNLIMITED

Procurement Executive, Ministry of Defence
Farnborough, Hants

REPORT DOCUMENTATION PAGE

Overall security classification of this page

UNLIMITED

As far as possible this page should contain only unclassified information. If it is necessary to enter classified information, the box above must be marked to indicate the classification, e.g. Restricted, Confidential or Secret.

1. DRIC Reference (to be added by DRIC)	2. Originator's Reference	3. Agency Reference	4. Report Security Classification/Marking
	RAE TR 81155	N/A	UNLIMITED

5. DRIC Code for Originator	6. Originator (Corporate Author) Name and Location
7673000W	Royal Aircraft Establishment, Farnborough, Hants, UK
5a. Sponsoring Agency's Code	6a. Sponsoring Agency (Contract Authority) Name and Location
N/A	N/A

7. Title: Wave drag from oblique cross-sectional areas

7a. (For Translations) Title in Foreign Language

7b. (For Conference Papers) Title, Place and Date of Conference

8. Author 1. Surname, Initials	9a. Author 2	9b. Authors 3, 4	10. Date	Pages	Refs.
Pike, J.			January 1982	35	12

11. Contract Number	12. Period	13. Project	14. Other Reference Nos.
N/A	N/A		Aero 3522

15. Distribution statement
(a) Controlled by –
(b) Special limitations (if any) –

16. Descriptors (Keywords) (Descriptors marked * are selected from TEST)

Wave drag*. Area rule.

17. Abstract

A simple numerical method for estimating the oblique cross-sectional areas of general body shapes is described together with the computer coding. These areas are used in conjunction with Eminton's method for evaluating supersonic wave drag, to produce some accurate estimates of the wave drag of cone-cylinders and some rather less accurate estimates for smooth axisymmetric bodies.

UDC 533.6.013.128 : 533.696.34 : 533.696.5

ROYAL AIRCRAFT ESTABLISHMENT

Technical Report 81155

Received for printing 4 January 1982

WAVE DRAG FROM OBLIQUE CROSS-SECTIONAL AREAS

by

J. Pike

SUMMARY

A simple numerical method for estimating the oblique cross-sectional areas of general body shapes is described together with the computer coding. These areas are used in conjunction with Eminton's method for evaluating supersonic wave drag, to produce some accurate estimates of the wave drag of cone-cylinders and some rather less accurate estimates for smooth axisymmetric bodies.

Departmental Reference: Aero 3522

LIST OF CONTENTS

1 INTRODUCTION

The wave drag of slender pointed bodies of general cross-section has been shown[1], within the restrictive assumptions of slender-body theory, to be given by

$$D = -4\pi\rho \int_0^{\ell}\int_0^{\ell} f'(x)f'(s)\,\log|x-s|\,dxds$$

$$+ 4\pi\rho f(\ell) \int_0^{\ell} f'(x)\,\log(\ell - x)\,dx$$

$$- 16\pi^2\rho U^2 \oint_c \phi\,\frac{\partial\phi}{\partial\nu}\,d\sigma + O(\tau^6 \log\,\tau) \qquad (1)$$

where x is the streamwise axis, f is the strength of a smooth source distribution along the x axis (for $0 \leqslant x \leqslant \ell$), ℓ is the body length, τ is a thickness parameter and ρ and U are the free-stream density and velocity respectively. The third term in equation (1) is a contour integral round the base of the body, where ϕ is the slender-body potential and ν is the normal to the contour in the base plane.

The source distribution f can be related to the cross-sectional area of the body by the simple relation

$$f(x) = -\frac{U}{4\pi} S'(x) \qquad (2)$$

permitting the wave drag of equation (1) to be expressed directly in terms of the geometry. For non-lifting configurations which have a pointed base or which merge smoothly into a cylinder parallel to the free stream, the second and third terms on the right-hand side of equation (1) are zero, and equations (1) and (2) reduce to the more familiar Von Karman wave drag integral form:

$$\frac{D}{q} = -\frac{1}{2\pi} \int_0^{\ell}\int_0^{\ell} S''(x)S''(s)\,\log|x-s|\,dxds\ . \qquad (3)$$

The first practical application of equation (3) to non-slender bodies was made by Whitcomb[2], who observed that, at transonic speeds, the drag of a wing-body combination was approximately the same as that of an equivalent body of revolution with the same cross-sectional area distribution. Although Whitcomb stated his area rule in terms of areas, for theoretical justification it is necessary to consider the body in terms of source distributions. The equivalent axial body is then one which has a source distribution formed from the sum of the sources lying in planes perpendicular to the free-stream direction (which are approximately Mach planes in transonic flow).

When the Mach number is not close to 1, the Mach lines are not nearly perpendicular to the stream and some modification to the area rule is required. In supersonic flow the equivalent source strength is taken as the sum of the sources lying on the Mach plane

$$x - \beta y \sin\theta - \beta z \cos\theta = x_0 \qquad (4)$$

where θ (the cylindrical polar angle) and x_0 (the intersection of the plane with the streamwise body axis), are parameters which define the particular Mach plane. The wave drag of a non-lifting body can then be expressed as[3]

$$\frac{D}{q} = -\frac{2\rho}{q}\int_0^{2\pi}\int_0^{\ell(\theta)}\int_0^{\ell(\theta)} f'(x,\theta)f'(s,\theta)\log|x-s|\,dx\,ds\,d\theta \quad . \qquad (5)$$

The evaluation of this triple integral can be simplified if the wing sources (f_w) and the fuselage sources (f_b) are separated. That is, writing $f'(x,\theta)$ as $f'_b + f'_w$, equation (5) splits into terms representing the drag of the fuselage alone, the drag of the wing alone and the wing-fuselage interference drag. If the fuselage is slender, the drag of the fuselage can be obtained from equation (3) using normal cross-sectional areas (which do not vary with θ), and also the order of the integration of the interference term can be reversed allowing the θ integration to be simplified. If the wing is described adequately by sources lying in a plane, the expression for the drag of the wing alone can also be simplified analytically, although for the wing in isolation a better estimate of the drag may often be obtained by empirical methods.

When applied to aircraft which really are slender (such as Concorde) the above techniques give good estimates of the wave drag[4]. More recently, however, various attempts to estimate the wave drag of fighter-type aircraft have had mixed success, with sometimes significant differences occurring in the estimates from methods which only appear to differ in the details of their application. It is to be expected that the application of the methods to aircraft which are not slender would lead to decreased accuracy; what has proved to be disappointing, however, is the lack of pattern in the error and the consequent limited success of modifications intended to reduce this error.

One potential source of difficulty for some methods is the treatment of the fuselage and wing by differing techniques (namely normal and oblique cuts respectively). Modern fighter aircraft often have the root of the wing extended forward along the fuselage or blended into the fuselage, so that it is not clear how the components should be distinguished. A conceptually simpler technique, which removes this difficulty, is to generate the area distributions for the triple integral using 'all oblique cuts', that is, to obtain the cross-sectional areas from the intersection of the wing and fuselage with the Mach planes given by equation (4), and to evaluate the triple integral directly.

Which of the two approaches is the more applicable to less slender bodies is unclear, but the latter approach would appear to be at least worthy of investigation. The reason

for its neglect has been mechanical rather than theoretical, in that it has proved to be surprisingly difficult to find the required oblique cross-sectional areas of even quite simple bodies.

In section 2 of the Report a novel numerical technique is presented for obtaining oblique cross-sectional areas for general body shapes and in section 3 the evaluation of wave drag from the lengthwise distribution of these areas using Eminton's[5] method is briefly described. A computer program listing with subroutines for several classes of bodies is given in the Appendix (examples from these classes are used in section 4), together with a description of the input parameters. The accuracy of prediction of wave drag for various bodies is discussed in section 4.

2 BODY DESCRIPTION

The numerical description of the body needs to be both efficient in its use of computer storage and of such a form that oblique cross-section areas may easily be found. In this section a body description meeting both these requirements is developed.

Consider a line parallel to a convenient body axis, x, which intersects the body as shown in Fig 1. The line is uniquely identified by its y and z coordinates (y_0 and z_0 say) and the section of the line inside the body is defined by the x values (x_i, with $x_{i+1} > x_i$) at its intersections with the body. For most simple bodies two x values are sufficient (x_1 and x_2), but for more complex bodies multiple intersections may occur requiring further x_i values.

Using a slender-body analogy, the body volume local to the line y_0, z_0 is considered to be concentrated on that line. The area of an oblique cross-section of this element can then immediately be related to the normal cross-section area as in slender-body theory. Thus if $A(x)$ is the elemental cross-sectional area associated with y_0, z_0 (corresponding to S used in equation (2) for slender body theory), then close to the intersection points x_i, A will vary according to the slope of the body, and remote from these points it will be constant. If the constant cross-section area is A_{max} inside the body then the element can be adequately defined by y_0, z_0, x_i, A_{max} and the variation of A near the intersection points x_i.

To achieve a compact representation of the body in the computer it is necessary to reduce to a minimum the number of values defining each element. This is achieved here by considering the computer representation and the physical representation of the data together. In the computer it is natural to locate the numbers defining the elements in rectangular arrays. The y_0 and z_0 arrays can be discarded if the physical position of the elements in space can be simply related to the position of numbers in the arrays which contain x_i and the other numbers necessary to describe the element. The simplest mapping in which the y_0, z_0 distribution is a rectangular array of points in physical space is wasteful, because some of the (y_0,z_0) lines may not intersect the body. In Fig 2, showing a front view of the body, it is indicated how the physical representation of a typical column of z_0 points is compressed such that nearly all the lines (y_0,z_0) intersect the body. The extreme points (denoted by $I_z = 1$ and $I_z = N_z$) are arranged to lie within $\Delta z/2$ of the surface, where Δz is the distance between adjacent points and is

independent of z. Note that z_{min} and z_{max} represent the outer limits of the elemental area representation (rather than the z coordinates of the end points of the column) such that

$$z_{max} - z_{min} = N_z \Delta z \tag{6}$$

and

$$z_0(I_z) = z_{min} + (I_z - 0.5)\Delta z \qquad I_z = 1,2,\ldots,N_z \; . \tag{7}$$

In this mapping the total number of points in each column is kept constant (at N_z) but the z_{min}, z_{max} and Δz values vary from column to column. The y spacing of the columns of y_0, z_0 values is assumed to be constant, so that the interval between the columns, Δy, is constant.

The body is also assumed to be symmetric about the plane $y = 0$ such that

$$y_0(I_y) = (I_y - 0.5)\Delta y \qquad I_y = 1,2,\ldots,N_y \tag{8}$$

and

$$\Delta y = y_{max}/N_y \; . \tag{9}$$

With this representation, the value of A_{max} is given by $\Delta y \Delta z$ and the definition of an element needs only its x_i values and information on the variation of A near the intersection points ($x = x_i$). The variation of A near the intersection points depends on the surface slope, which can be approximated from the values of x_i for the adjacent elements. Because the physical array of (y_0,z_0) points is not rectangular, the only immediately useful adjacent x_i values in the array are those in the same column (*ie* in the adjacent z positions); the x_i values in adjacent y positions do not correspond to the same value of z and use of them would require interpolation. There is an additional requirement on the variation of A with x, in that the elemental bodies are assumed to be slender, requiring an upper limit on the permitted value of dA/dx. With these restrictions, the method used to approximate dA/dx is shown diagramatically in Fig 3. The variation in the area distribution A between $\Delta y \Delta x$ and zero is taken to be the normal cross-sectional area of a tapered wedge with a leading edge lying in a vertical plane y = constant, and oblique to the x-axis. The wedge is tapered over a distance of $2\Delta y$ to satisfy the slenderness requirement and consequently the maximum value of dA/dx is $\Delta z/2$. The angle of the slant of the leading edge of the wedge is obtained from the x_i values of the adjacent z_0 points. This approximation to dA/dx is very easy to calculate and in practice proves to be an adequate representation of the change in the area. The representation of each element can now be described by its intersection points only, that is for simple bodies the entry and exit points x_1 and x_2.

In the computer these x values are conveniently stored in a three-dimensional array, which in the program as listed in the Appendix is called the XBM array. The size of this array is N_y, N_z, IB where IB is the maximum number of intersections (*ie* the

maximum value of i). The computation of oblique cross-sectional areas with this representation is now no more complicated than that of normal cross-sectional areas, for it is necessary only to sum the contribution from each element at the x value where its centreline is cut by the plane of equation (4). To find an oblique cross-sectional area where the normal to the cutting plane makes an angle ψ with the element axis, the contribution from each element is either zero or $\Delta y \Delta z \sec \psi$, except for the elements cut near their end points when the correction due to taper is applied. This simple repetitive process is ideally suited to solution on a computer.

3 EVALUATION OF THE WAVE DRAG

Given the ability to calculate oblique cross-section areas at specific x values, the triple-integral of equation (5) can be evaluated to give the wave drag by using Eminton's method[5]. This method depends on representing the area distribution as a summation of sine waves (similar to a Fourier series), which on substitution in the integrand of the triple-integral gives terms which can be integrated analytically with respect to x. The method has been widely used and is not discussed further here.

The computer evaluation is performed using the Fortran subroutine WAVEDR listed with the main program in the Appendix. As this subroutine may usefully be employed other than in the present program it is written to be independent of the main program. Its only external reference is to the subroutine INVERT, which is a subroutine to invert a matrix A onto itself. The values required in the evaluation are passed through to parameter list. The cross-sectional areas to be used in the calculation are contained in the vector AREA(NX). The x spacing of the areas (Δx) is given by DX in the program, and on exit from the subroutine the parameter WDRAG has the value $D(\theta)/q$ in the units of AREA. In the main program, WAVEDR is used to obtain D/q at NTHETA values of θ between (and including) $\theta = 0$ and $\theta = \pi$. The body wave drag is then taken as the appropriate mean of these values, that is with contribution from the end values at $\theta = 0$ and $\theta = \pi$ halved and the mean divided by π.

4 ASSESSMENT OF THE METHOD

Wave drag estimates from the supersonic area rule method discussed here are compared with more accurate wave drag estimates to validate the computer program and to assess the error involved in using supersonic area rule to estimate the drag of non-slender bodies.

4.1 Comparison with slender body theory

To investigate the errors introduced by the approximation to the body geometry and the other approximations used in the present method, the wave drag of a range of Sears-Haack[6,7] bodies and some non-axisymmetric bodies is computed and compared with the values obtained analytically using slender-body theory.

(a) Sears-Haack bodies

The shape of a Sears-Haack body of unit length is given by[7]

$$S(x) = \pi R^2 = \frac{128V}{3\pi} x^{\frac{3}{2}} (1 - x)^{\frac{3}{2}} \quad . \tag{10}$$

The wave drag of this body according to slender body theory is D ,

where
$$\frac{D}{q} = \frac{128V^2}{\pi} \quad . \tag{11}$$

Eliminating the volume V between equations (10) and (11) we obtain an equation for the body shape with the wave drag as a parameter, *ie*

$$x = \frac{1}{2} \pm \frac{1}{2}\left(1 - \left\{\frac{4.5\pi^3 R^4}{(D/q)}\right\}^{\frac{1}{3}}\right)^{\frac{1}{2}} \quad . \tag{12}$$

The substitution of a value of D/q in this equation defines the geometry of a test body with this theoretical value of D/q . Thus for this body, the computer program (with $M_\infty = 1$) ought to recover the prescribed value of D/q . The variation in the accuracy of the program with the thickness of the body and the number of elements (N_y, N_z) defining it is shown in Figs 4 and 5 respectively. In Fig 4 the error obtained when estimating the wave drag of a Sears-Haack body represented by 2500 elements (*ie* $N_y = N_z = 50$) is shown. For a range of D/q from zero to 0.1 (representing a body thickness to length ratio from 0 to 0.327) we can see that the numerically derived value is within $+\frac{1}{2}\%$ of the theoretical value. The large range of body thickness represented can be appreciated from the body side views shown for D/q equal to 0.02 and 0.08. The pattern of fluctuation in the error is of little significance because it depends on the particular relationship between the elements and the cross-sectional cuts. As the number of elements is increased it is shown in Fig 5 how the accuracy increases, as might be expected, like $1/(N_y N_z)$.

(b) Norman's 'single bulge' body

A non-axisymmetric body for which oblique cross-section areas have been derived analytically[8] is illustrated in Fig 6. The body is symmetrical about the plane $y = 0$ and is described by Norman as a 'single bulge' body. The body cross-section for $x \leq 0$ (*ie* upstream) is square. Between $x = 0$ and $x = 6$ the body cross-section changes smoothly to an 'inverted T' section in such a manner that the normal cross-section area of the body remains constant. Downstream of this change (*ie* $x > 6$) the cross-section shape is again constant. The body is defined mathematically by

Top surface	$z = 1.5$	(13)
Bottom surface	$z = -1.5$	(14)
Sides $(z > 0)$	$y = \pm(1.5 - H(x))$	(15)
Sides $(z < 0)$	$y = \pm(1.5 + H(x))$	(16)

with sufficient of the plane $z = 0$ to complete the surface. The function $H(x)$ is defined by

$H(x) = 0$	$x \leq 0$	(17)
$H(x) = \frac{1}{2} - \frac{1}{2}\cos(\pi x/6)$	$0 \leq x \leq 6$	(18)
$H(x) = 1$	$x \geq 6$.	(19)

This body critically tests the accuracy of the present method because of the large areas of high slope (*ie* dy/dx) on its vertical slab sides. As the cross-sectional area of the body is constant, the wave drag using normal cross-sectional areas in equation (3) is zero. The wave drag from oblique areas will not be zero, however. The actual values of the drag coefficient found from the analytically derived oblique areas[8] are shown in Fig 7.

By introducing the supersonic-hypersonic similarity parameter $\tau(M^2 - 1)^{\frac{1}{2}}$, Fig 7 can be made to represent the wave drag of the class of bodies formed by a linear stetching of the body defined by Norman along the x axis. The function H(x) is then defined by

$$H(x) = 0 \qquad x \leqslant 0 \tag{20}$$

$$H(x) = \tfrac{1}{2} - \tfrac{1}{2}\cos(\pi x/6\tau) \qquad 0 \leqslant x \leqslant 6/\tau \tag{21}$$

$$H(x) = 1 \qquad x \geqslant 6/\tau \tag{22}$$

where unit value of τ recovers Norman's body as described by equations (13) to (19).

The difference in the wave drag between the present method and that shown in Fig 7, is indicated in Fig 8 for M = 1.4 and M = 2.0. To obtain a 1% accuracy it is necessary to use much greater detail in the body specification than for the Sears-Haark body, values of 200 being required for N_y and N_z. The reasons for this are twofold. First the body has slab sides with large gradients near $x = 3/\tau$ which require detailed representation. The rod geometry described in Fig 3 for the area distribution at the ends of the rod is only partially effective for vertical slab sides. This makes the geometry of this particular body a good test of the method and accounts for the higher values of N_y. The second reason is that the body has a constant cross-sectional area normal to the flow direction and hence the wave drag is due to interference effects which only appear when $M > 1$. This interference drag is much smaller than the drag normally associated with the thickness, for example, the drag coefficient[8] at M = 1 of the top and bottom halves of the body taken separately is 0.138, which is more than three times the maximum value shown in Fig 7. Thus a better representation of the body shape than for a closed axisymmetric body is required to obtain the 1% accuracy.

4.2 Comparison with wave drag from other methods

To determine whether the use of oblique cuts has any advantages in accuracy over other linear methods, it is necessary to estimate the wave drag of bodies whose wave drag is known accurately.

(a) Cone-cylinder bodies

Consider axisymmetric cone-cylinder bodies where the cylinder extends downstream to infinity. The wave drag of a cone-cylinder with the cross-section area of the cylinder as reference area, is the same as the constant pressure coefficient on the cone[9]. In Fig 9 the wave drag of cone-cylinders with 10° and 15° semi-vertex angles is shown for a range of Mach numbers, obtained from exact theory and several approximate theories[10]. The exact result is given by the continuous line. The results from slender body theory are shown by the fine dotted line and those for linear theory using the exact

boundary conditions on the cone by the dot-dash line. Both of these theories are clearly inadequate for this case. A second-order approximation due to Broderick[11] and shown by the plain dashed line is rather better, but even that gives an error of over 10% for the 15^{o} cone above M = 1.5.

The Supersonic Area Rule (SAR) method described here gives the 'cross' line which is within 5% of the exact value for the 10^{o} cone and for the 15^{o} cone up to M = 2.5. This is considerably better than the other methods shown.

(b) Hewitt's axisymmetric body

Recently[12] the wave drag of a particular axisymmetric body has been calculated using both the method of characteristics and the exact solution to the linear equation. The radius of the body in terms of the streamwise coordinate x , is given by

$$R = \tfrac{1}{2}x(1 - x)^2 \qquad 0 \leqslant x \leqslant 1 \qquad (23)$$

where the body is assumed to be of unit length and has a cusp at its downstream end (*ie* at x = 1). The shape of the body and the wave drag from both the method of characteristics (continuous line) and the exact solution of the linear equation (dotted line) are shown in Fig 10. The good agreement between these values is not matched by the supersonic area rule estimate, as shown by the dashed line for $N_y = N_z = 100$.

We saw in the case of the cone-cylinder that the supersonic area rule estimate was close to the exact value for cone angles of less than about half the Mach angle, but that large over-estimates of the drag were likely to occur for larger cone angles. For $M \geqslant 1.6$ the nose angle of the body of equation (23) is more than 2/3 of the Mach angle, and hence a high wave drag estimate is not necessarily in conflict with the cone-cylinder comparison. A more detailed investigation of the error is necessary therefore, to establish whether the smooth axisymmetric bodies of equation (23) have an error pattern consistent with that of cone-cylinder bodies.

First, consider the wave drag of bodies which are described by equation (23), but which are truncated such that the base is not at x = 1 but at some point $0 \leqslant x_L \leqslant 1$. Thus when x_L is small the body resembles a cone-cylinder body and when x = 1 the body is that defined by equation (23). The wave drag at M = 1.6 for these bodies is shown in Fig 11, both for SAR and the method of characteristics, and it can be seen that the large differences between them found at $x_L = 1$ in Fig 10, occur for most of the range of x_L . The error in the SAR estimate, when expressed as a percentage of the method of characteristics value is shown in Fig 12. The variation in the error over the forebody and the rather higher value established over the rest of the body suggests that the large nose angle only partially explains the drag error.

Cone-cylinder results would suggest that the error in the wave drag should decrease rapidly as the body becomes more slender. In Fig 13 the wave drag for bodies of unit length whose radius is given by

$$R = \lambda x(1 - x)^2 \qquad 0 \leqslant x \leqslant 1 \qquad (24)$$

are shown. As expected the error decreases as the body becomes more slender (λ smaller), but even for $\lambda < 0.35$, when the nose angle is less than half the Mach angle, errors of 15% in the wave drag estimate using SAR still occur. This is, however, significantly better than the estimate using normal cuts, which is shown in Fig 13 to have an error of nearly 25% when $\lambda = 0.35$.

The lack of consistency in the results even for these comparatively simple axisymmetric shapes indicates that comparisons are needed for a much wider range of shapes before any more general conclusions can be reached. The encouraging factor in the examples shown is the significant improvement in the drag estimates from the 'oblique cuts' method used here when compared with the estimates from 'normal cuts'. It is hoped that the provision of a simple technique for obtaining oblique cross-sectional areas will encourage wave drag estimates based on oblique cuts to be included in future wave drag comparison studies, enabling the strengths and weaknesses of this method to be compared with those of other methods.

5 CONCLUSIONS

A new method of obtaining oblique cross-sectional areas of general body shapes is shown to be both efficient and accurate. When used in conjunction with Eminton's method for evaluating the slender-body wave drag integral, analytical slender-body results are reproduced within 1%. The method is used to estimate the wave drag of cone-cylinder bodies and smooth axisymmetric bodies for which the exact wave drag is known from the method of characteristics with a 'fitted' bow shock wave. For cone-cylinders with nose semi-angles up to half the Mach angle, the estimates of the wave drag were found to be within 5% of the exact values. This standard of accuracy was not maintained for the smooth axisymmetric bodies, however, when typically over-estimates of 15% occurred for nose semi-angles equal to half the Mach angle. Even this result represents a significant improvement in the accuracy obtained using normal cuts, but results for a larger range of bodies are needed before any general conclusions can be drawn.

Appendix

THE SPECIFICATION OF BODY SHAPE AND CALCULATION DATA FOR THE WAVE DRAG PROGRAM

The program is written for easy and convenient use. The body is defined by the Fortran FUNCTION X(Y,Z,IB). This body function will in general be supplied by the user and must give the X (streamwise) value of the body surface for any given Y and Z value (with Y = 0 a plane of symmetry). Often there will be more than one value of X for any Y and Z value, that is one value of X for each time this line enters or leaves the body. IB defines which intersection is required, such that IB = 1 is the first (most upstream) body entry X value and further intersections are taken in ascending X value. For a simple closed body X will have significant values only when IB is 1 or 2. An annotated listing of the main program and the X function used appears at the end of this Appendix. A more generally applicable X function for bodies of form F(x,y,z) = 0 is also listed.

For convenience, the run parameters are defined in the FUNCTION DATA(N). These are described below and examples of typical DATA functions are included in the program listing.

RM	Mach number (>1)
Z0	Z coordinate of the X line about which the wave drag is calculated (value usually zero)
NBODY	A reference body number for users' convenience
NOPDEV	Output channel for results
NOPCTL	Controls output of results Value 1 Run parameters and wave drag only 2 Wave drag for each θ value given also 3 Areas for each θ and DX given also 4 Body projection and X0 limits given also
NX	Number of oblique cuts for each value of θ . ($NX \leq 100$)
NY	Number of points across body in Y direction ($NY \leq 100$ but can be increased by altering DIMENSION statement in main program)
NZ	Number of points across body in Z direction ($NZ \leq 100$ but can be increased by altering DIMENSION statement in main program)
NTHETA	Number of θ values from 0 to π at which wave drag is calculated ($NTHETA \leq 100$ but can be increased by altering DIMENSION statement in main program)
NB	Maximum number of times any X line enters or leaves the body (if $NB > 2$ alter DIMENSION statement in main program)

```
C MAIN PROGRAM.
C
C
C  PROGRAM WAVEDRAG.
C  CALCULATES THE WAVEDRAG OF AN ANALYTICALLY DEFINED BODY X=X(Y,Z)
C  USING SUPERSONIC AREA RULE WITH ALL OBLIQUE CUTS.
C ARRAYS XBM(NY,NZ,NB),AREA(NX),WDRAG(NTHETA),ZMINV(NY),ZMAXV(NY),DZ(NY)
C  MAX VALUES OF NY,NZ,NB,NTHETA SET BY DIMENSION & COMMON FOLLOWING.
      DIMENSION XBM(100,100,2),AREA(100),WDRAG(100),DZ(100)
      COMMON/A/ ZMINV(100),ZMAXV(100)
C ZMINV & ZMAXV ARE ONLY IN COMMON FOR THE SIZE CHECK OF ZMINV >NY BELOW
C
C  GET RUN DATA.
      RM=DATA(1)
      Z0=DATA(2)
      NOPDEV=DATA(3)+0.1
      NOPCTL=DATA(4)+0.1
      NBODY=DATA(5)+0.1
      NX=DATA(6)+0.1
      NY=DATA(7)+0.1
      NZ=DATA(8)+0.1
      NTHETA=DATA(9)+0.1
      NB=DATA(10)+0.1
C
C CHECK NY SIZE AGAINST ZMINV SIZE ALLOTTED IN COMMON STATEMENT.
      DO 1990 N=1,NY
 1990 ZMINV(N)=0
      ZMAXV(1)=99999
      DO 2000 N=2,NY
 2000 IF(ZMINV(N).GT.99998.99.AND.ZMINV(N).LT.99999.01)GOTO 2010
      GOTO2030
 2010 WRITE(NOPDEV,2020)
 2020 FORMAT(' NY BIGGER THAN ARRAY SPACE GIVEN IN DIMENSION STATEMENT'
      WRITE(NOPDEV,110)NX,NY,NZ,NTHETA,NB
      GOTO 1100
 2030 CONTINUE
C
C  DESCRIBE THE BODY PROJECTION SHAPE (ON A PLANE X=CONST.)
C  INITIAL GUESS
      YMAX=0.1
      ZMIN=-0.1
      ZMAX=0.1
C
C  FIRST EXPAND YMAX,ZMIN & ZMAX TO CONTAIN PROJECTION.
   1  YMAX0=YMAX
      ZMIN0=ZMIN
      ZMAX0=ZMAX
      DO 2 IY=1,NY
      Y=(IY-0.5)*YMAX/NY
      IF(X(Y,ZMAX,1).LT.9000)ZMAX=1.2*ZMAX-0.2*ZMIN
      IF(X(Y,ZMIN,1).LT.9000)ZMIN=1.2*ZMIN-0.2*ZMAX0
      IF(ZMAX-ZMAX0+ZMIN0-ZMIN.GT.(ZMAX-ZMIN)/20)GOTO 1
   2  CONTINUE
      DO 3 IZ=1,NZ
      Z=ZMIN+(IZ-0.5)*(ZMAX-ZMIN)/NZ
      IF(X(YMAX,Z,1).LT.9000)YMAX=1.2*YMAX
      IF(YMAX-YMAX0.GT.YMAX/20)GOTO 1
   3  CONTINUE
```

```
C
C   NOW REDUCE YMAX,ZMIN,ZMAX TO HUG PROJECTION
C   NEW ZMIN & ZMAX VECTORS IN ZMINV(NY),ZMAXV(NY)

      DZ0=(ZMAX-ZMIN)/NZ/5
      Y=YMAX
    4 YMAX=Y
      DO 7 IY=1,NY
      ZMAX=ZMAX0
      ZMIN=ZMIN0
      IIY=NY+1-IY
      Y=(IIY-0.5)*YMAX/NY
      Z=ZMAX+DZ0/2
    5 Z=Z-DZ0
      IF(Z.LE.ZMIN)GOTO 4
      IF(X(Y,Z,1).GT.9000)GOTO 5
      ZMAXV(IIY)=AMIN1(ZMAX,Z+DZ0)
      Z=ZMIN-DZ0/2
    6 Z=Z+DZ0
      IF(X(Y,Z,1).GT.9000)GOTO 6
      ZMINV(IIY)=AMAX1(ZMIN,Z-DZ0)
    7 CONTINUE
C
C   SET UP DY,DZ,DTHETA & A GUESS AT XSTART,XFINIS.
      XS=9000
      XF=-9000
      IF(ABS(X(0,0,1)).LT.9000)XS=X(0,0,1)
      IF(ABS(X(0,0,NB)).LT.9000)XF=X(0,0,NB)
      DY=YMAX/NY
      DO 8 IY=1,NY
    8 DZ(IY)=(ZMAXV(IY)-ZMINV(IY))/NZ
      DTHETA=3.14159/(NTHETA-1)
C
C SET UP X VALUE ARRAYS (FOR PROJECTED SHAPE)
      DO 10 IB=1,NB
      DO 10 IY=1,NY
      DO 10 IZ=1,NZ
      XBM(IY,IZ,IB)=20000
      Y=(IY-0.5)*DY
      Z=ZMINV(IY)+(IZ-0.5)*DZ(IY)
      IIB=IB
      XVALUE=X(Y,Z,IIB)
      XBM(IY,IZ,IB)=XVALUE
      IF(ABS(XVALUE).GT.9000)GOTO 10
C   FIND UPSTREAM(XSTART) & DOWNSTREAM(XFINIS) POINTS ON REF AXIS(Y=0,Z=Z0)
C   FOR WHICH D(OBLIQUE AREA)/DX=0.
      RBETA=SQRT((Y*Y+(Z-Z0)*(Z-Z0))*(RM*RM-1))
      XS=AMIN1(XS,XVALUE-RBETA)
      XF=AMAX1(XF,XVALUE+RBETA)
   10 CONTINUE
      XSTART=XS-(XF-XS)/NX
      XFINIS=XF+(XF-XS)/NX
      DX=(XFINIS-XSTART)/(NX-1.)
```

```
C
C  WRITE RUN DATA.
      WRITE(NOPDEV,100)RM,Z0,NBODY,NOPDEV,NOPCTL
  100 FORMAT(/' DATA',/,' MACH NUMBER(RM)=',F6.4,', AXIS Z(Z0)=',F5.3,/,
     &' NBODY=',I2,' NOPDEV=',I2,', NOPCTL=',I2)
      WRITE(NOPDEV,110)NX,NY,NZ,NTHETA,NB
  110 FORMAT(' NX=',I3,', NY=',I3,', NZ=',I3,', NTHETA=' ,I3,', NB=',I3)
      IF(NOPCTL.LT.4)GOTO 20
      WRITE(NOPDEV,120)XSTART,XFINIS,YMAX
  120 FORMAT(/' XSTART=',F8.3,',  XFINIS=',F8.3,',  YMAX=',F8.3)
      WRITE(NOPDEV,130)
  130 FORMAT('  ZMINV,ZMAXV')
      WRITE(NOPDEV,140)(ZMINV(J),J=1,NY)
      WRITE(NOPDEV,140)(ZMAXV(J),J=1,NY)
  140 FORMAT(1X,10F7.4)
   20 CONTINUE
C
C
C  EVERYTHING IS NOW SET UP SO.........
C  FIND OBLIQUE AREAS & WAVEDRAG FOR EACH THETA.
C
C MAIN LOOP (THETA)
      DO 60 ITHETA=1,NTHETA
      THETA=DTHETA*(ITHETA-1)
      BS=SQRT(RM*RM-1)*SIN(THETA)
      BC=SQRT(RM*RM-1)*COS(THETA)
C
C 2ND LOOP (X)
      DO 50 IX=1,NX
      X0=XSTART+(IX-1)*DX
      SUMKUT=0
C
C 3RD LOOP (Y)
      NYY=1-NY
      DO 40 IY=NYY,NY
      IIY=(IABS(2*IY-1)+1)/2
      Y=(IY-0.5)*DY
      PARTYZ=X0+BS*Y+BC*(ZMINV(IIY)-DZ(IIY)/2-Z0)
      PARTZ=BC*DZ(IIY)
C
C 4TH LOOP (B)
      DO 40 IB=1,NB
      DZIIY=(2*MOD(IB,2)-1)*DZ(IIY)
      KUTOLD=0
      CUTOLD=20000
C
C 5TH LOOP (Z)
      DO 40 IZ=1,NZ
      KUT=0
      XCUT=PARTYZ+IZ*PARTZ-XBM(IIY,IZ,IB)
      IF(XCUT.GT.0)KUT=1
      XCUTS=AMIN1(1.0,AMAX1(0.0,XCUT/DY/2.0+0.5))
      SUMKUT=SUMKUT+XCUTS*DZIIY
      IF(KUT.EQ.KUTOLD)GOTO 30
```

```
C
C INTERPOLATE IN Y PLANE IF SURFACE POINT.
      XCUTY=XCUT
      XCUTOY=CUTOLD
      DENOM=ABS(XCUT)+ABS(CUTOLD)
      IF(DENOM.LT.10000)GOTO 25
C  EXTRAPOLATE TO EDGE OF NON-POINTED BODY.
      IF(ABS(CUTOLD).GT.10000.AND.ABS(XCUT).GT.10000)GOTO 25
      PART=PARTYZ+IZ*PARTZ-2*XBM(IIY,IZ,IB)
      IF(ABS(CUTOLD).GT.10000)XCUTOY=PART+XBM(IIY,IZ+1,IB)
      IF(ABS(XCUT).GT.10000)XCUTY=PART+XBM(IIY,IZ-1,IB)
      DENOM=ABS(XCUTY)+ABS(XCUTOY)
  25  SUMKUT=SUMKUT+(XCUTY+XCUTOY)/DENOM/2*DZIIY
  30  KUTOLD=KUT
      CUTOLD=XCUT
  40  CONTINUE
C
C NOW CALC. AREA & FROM AREAS GET WAVE DRAG FOR THIS THETA
      AREA(IX)=SUMKUT*DY
  50  CONTINUE
      CALL WAVEDR(DX,AREA,NX,WDRAG(ITHETA))
      THETAD=THETA*180/3.14159
      IF(MOD(ITHETA-1,1).NE.0.AND.ITHETA.NE.NTHETA)GOTO 60
      IF(NOPCTL.LT.2)GOTO 60
      WRITE(NOPDEV,200)THETAD,WDRAG(ITHETA)
 200  FORMAT(/' THETA=',F6.2,',   WAVEDRAG=',F12.6)
      IF(NOPCTL.LT.3)GOTO 60
      WRITE(NOPDEV,210)
 210  FORMAT(' NX AREAS')
      WRITE(NOPDEV,220)(AREA(J),J=1,NX)
 220  FORMAT(1X,10F7.4)
  60  CONTINUE
C
C AVERAGE WAVEDRAG OVER THETA (AWDRAG)
      WDRAG(1)=WDRAG(1)/2
      WDRAG(NTHETA)=WDRAG(NTHETA)/2
      AWDRAG=0
      DO 70 ITHETA=1,NTHETA
  70  AWDRAG=AWDRAG+WDRAG(ITHETA)/(NTHETA-1)
      WRITE(NOPDEV,300)AWDRAG
 300  FORMAT(//' BODY WAVEDRAG=',F13.7,'   (SQ. LENGTH UNITS OF X,Y,Z)')
1100  CONTINUE
      STOP
      END
```

```
      SUBROUTINE WAVEDR(DX,AREA,NX,WDRAG)
C  NX AREAS ARE SUPPLIED IN AREA,SPACED AT INTERVALS DX
C  WDRAG IS PUT EQUAL TO THE WAVE DRAG (D/Q) USING EMINTONS METHOD
      DIMENSION AREA(NX),C(100),P(100,100),WORK1(100),WORK2(100)
      NX2=NX-2
      DO 10 IX=1,NX2
      X=IX/(NX-1.)
      U=(2*ATAN(SQRT(X/(1-X)))-2*(1-2*X)*SQRT(X-X*X))/3.14159
      C(IX)=AREA(IX+1)-AREA(1)-(AREA(NX)-AREA(1))*U
      DO 10 IY=1,NX2
      Y=IY/(NX-1.)
      XY1=X+Y-2*X*Y
      XY2=2*SQRT(X*Y*(1-X)*(1-Y))
      P(IX,IY)=XY1*XY2
      IF(IY.NE.IX)P(IX,IY)=-(X-Y)**2*ALOG((XY1+XY2)/(XY1-XY2))/2+XY1*XY2
 10   CONTINUE
C  INVERT THE ARRAY P(100,100) ONTO ITSELF
      CALL INVERT(P,NX2,100,WORK1,WORK2)
      SUM=0
      DO 20 IX=1,NX2
      DO 20 IY=1,NX2
 20   SUM=SUM+C(IX)*C(IY)*P(IX,IY)
      WDRAG=(4/3.1416*(AREA(NX)-AREA(1))**2+3.1416*SUM)/DX/DX/(NX-1)**2
      RETURN
      END
C
C
C
C
      SUBROUTINE INVERT(A,M,IA,IND,C)
C INVERTS AN M*M  MATRIX  'A' ONTO ITSELF,WHERE AI IS THE DECLARED
C  ARRAY SIZE AND IND(M) & C(M) ARE WORKING ARRAYS.
C
C LIBRARY SUBROUTINE INVERT CODE NOT INCLUDED.
```

```
C BODY & DATA FUNCTIONS FOR SEARS-HAACK BODY.
C
C
      FUNCTION X(Y,Z,IB)
C  SEARS-HAACK BODY OF GIVEN LENGTH (XL) & D/Q.
C  WHEN M=1.001 THIS TEST BODY SHOULD GIVE D/Q FROM SAR THE SAME AS
C  THAT ENTERED BELOW   (FOR ANY AOB & LENGTH )
C  NB=2 FOR THIS BODY.
C  RMAX=(2*L*L*DOQ/9/PI**3)**(1/4) PG283 NIELSEN,MISSILE AERODYNAMICS
C  AOB=RATIO A/B FOR ELLIPTIC CROSS-SECTION OPTION.
      X=20000
      AOB=1.0
      XL=1
      DOQ=0.08
      PI=3.14159
      TEMP=9*PI*PI*PI*((Y*Y/AOB+Z*Z*AOB)**2)/2/XL/XL/DOQ
      IF(TEMP.GT.1.OR.IB.GT.2)GOTO 10
      X=XL/2*(1-SQRT(1-TEMP**(1/3.)))
      IF(IB.EQ.2)X=XL-X
   10 RETURN
      END
C
C
C


      FUNCTION DATA(N)
      RM=1.0001
      Z0=0
      NOPDEV=5
      NOPCTL=4
      NBODY=1
      NX=25
      NY=50
      NZ=50
      NTHETA=3
      NB=2
      IF(N.EQ.1)DATA=RM
      IF(N.EQ.2)DATA=Z0
      IF(N.EQ.3)DATA=NOPDEV
      IF(N.EQ.4)DATA=NOPCTL
      IF(N.EQ.5)DATA=NBODY
      IF(N.EQ.6)DATA=NX
      IF(N.EQ.7)DATA=NY
      IF(N.EQ.8)DATA=NZ
      IF(N.EQ.9)DATA=NTHETA
      IF(N.EQ.10)DATA=NB
      END
```

```
C BODY & DATA FUNCTIONS FOR SINGLE BULGE BODY.
C
C
        FUNCTION X(Y,Z,IB)
C        SINGLE BULGE BODY (NB=2 FOR THIS BODY)
        X=20000
        IF(ABS(Z).GT.1.5)GOTO 999
        IF(ABS(Y).GE.2.5)GOTO 999
        IF(IB.GT.1)GOTO 10
        IF(ABS(Y).LE.1.5)X=-20000
        IF(X.LT.-9999)GOTO 999
        IF(Z.GE.0)X=6/3.14159*ACOS(4-2*Y)
        GOTO999
10      IF(ABS(Y).LE.0.5.OR.ABS(Y).GE.1.5.OR.Z.GE.0)GOTO 999
        X=6/3.14159*ACOS(2*Y-2)
999     RETURN
        END
C
C
C
C
      FUNCTION DATA(N)
      RM=1.4
      Z0=0
      NOPDEV=5
      NOPCTL=4
      NBODY=2
      NX=31
      NY=100
      NZ=100
      NTHETA=19
      NB=2
      IF(N.EQ.1)DATA=RM
      IF(N.EQ.2)DATA=Z0
      IF(N.EQ.3)DATA=NOPDEV
      IF(N.EQ.4)DATA=NOPCTL
      IF(N.EQ.5)DATA=NBODY
      IF(N.EQ.6)DATA=NX
      IF(N.EQ.7)DATA=NY
      IF(N.EQ.8)DATA=NZ
      IF(N.EQ.9)DATA=NTHETA
      IF(N.EQ.10)DATA=NB
      END
```

```
C  BODY FUNCTION FOR ANY BODY F(X,Y,Z)=0.
C
      FUNCTION X(Y,Z,IB)
C
C  X VALUE OF ANY FUNCTION F(X,Y,Z)=0 GIVEN Y & Z.
C  IB=1 SMALLEST X INTERSECTION VALUE (>XNOSE).IB=2 NEXT INTERSECTION
C
C  USER DEFINED FUNCTION F
C (BODY OF FIG 10 HERE).
      F(X,Y,Z)=(Y*Y+Z*Z)-0.25*X**2*(1-X)**4
C
C
C  USER SET RUN CONSTANTS
20    XERROR=0.000001
      XNOSE=0
      XBASE=1
C
   PROGRAM(NO USER CHANGE NEEDED).
C  INITIALISE
      IMAX=100
      XN=XNOSE
      XB=XBASE
      IBC=0
      FN=F(0.0,10000.0,10000.0)
C  FIND X FOR IB(X IN X TO X+DX HERE)
C  F CHANGES SIGN AS X CROSSES SURFACE.
100   DO 200 I=1,IMAX
      DX=(XB-XN)/(IMAX-1)
      X=XN+(I-1)*DX
      FOLD=FN
C  GET F VALUE
      FN=F(X,Y,Z)
110   IF(FN*FOLD.LE.0)IBC=IBC+1
       IF(IBC.EQ.IB)GOTO 300
200   CONTINUE
C  NO VALUE FOR THESE Y,Z,IB SO SET X=DEFAULT.
      X=20000
      GOTO 400
300   IF(X.LT.XN+0.000000001)X=-20000
      IF(X.LT.-19999)GOTO 400
C  CRAMP X-LIMITS XN&XB ONTO X
      XN=X-DX
      XB=X
      IBC=IBC-1
      FN=F(0.0,10000.0,10000.0)
      IF(DX.GT.XERROR*2)GOTO 100
      X=(XN+XB)/2.0
      IF(X.GT.XBASE)X=20000
      IF(X.LT.XNOSE)X=-20000
400   RETURN
      END
```

LIST OF SYMBOLS

A	cross-sectional area of body element
D	wave drag
f	source distribution function
H	body shape function (equations (17) to (22))
ℓ	body length (or streamwise length for which S varies)
M	Mach number
N_x	number of cross-sectional areas
N_y, N_z	number of body elements in y and z directions
q	kinetic pressure
R	body radius
S	body cross-sectional area (may be oblique)
V	free-stream velocity
V	body volume (equation (11))
x	streamwise coordinate
x_i	body element intersections with body surface
y, z	spanwise coordinates
y_0, z_0	coordinates of elemental body
β	$(M^2 - 1)^{\frac{1}{2}}$
Δx	interval between cross-sections
Δy, Δz	elemental body spacing
θ	cylindrical polar angle
ρ	free-stream density
σ, ν	coordinates parallel and normal to base contour (equation (1))
τ, λ	thickness parameters
ϕ	slender-body potential (equation (1))
ψ	angle between the normal to the cutting plane and the x axis ($\cos^{-1}(1/M)$)

REFERENCES

No.	Author	Title, etc
1	G.N. Ward	Linearised theory of steady high-speed flow. Cambridge Monographs on Mechanics and Applied Mathematics, Cambridge University Press (1955)
2	R.T. Whitcomb	A study of the zero-lift drag-rise characteristics of wing-body combinations near the speed of sound. NACA TR 1273 (1956)
3	J.N. Nielsen	Missile aerodynamics. McGraw Hill, New York, pp 283-300 (1960)
4	C.S. Leyman T. Markham	Prediction methods for aircraft aerodynamic characteristics. AGARD LS-67, paper 5, May 1974
5	E. Eminton	On the minimisation and numerical evaluations of wave drag. RAE Report Aero 2564 (ARC R & M 3341) (1958)
6	W.R. Sears	On projectiles of minimum wave drag. Quart. Applied Maths, Vol.14, No.4 (1947)
7	W. Haack	Geschossenformen kleinsten Wellenwiderstands, Ber. Lilienthal-Ges. Luftfahrt, Vol.139, pp 14-37 (1948)
8	G. Norman	Some calculations of the wave drag of bluff asymmetries. British Aerospace, Kingston - Brough Div, Note YAD 3342, November 1978
9	J.L. Sims	Tables of supersonic flow about right circular cones at zero angle of attack. NASA SP 3004 (1964)
10	M.J. Lighthill	Higher approximations in aerodynamic theory. Princeton University Press, p 126 (1960)
11	J.B. Broderick	Supersonic flow past a semi-infinite cone. Quart.J. Mech. and Appl. Math. 2, pp 121-128 (1949)
12	B.L. Hewitt	The calculation of wave drag in steady supersonic flow. British Aerospace, Warton Division, Report AE/A/651, July 1980

Fig 1



Fig 1 A typical body element

TR 81155



Fig 2 Front view (along x axis) showing a typical column of y_0, z_0 points

Fig 3



Fig 3 Cross-section area distribution of a typical rod



Fig 4 Error in estimating the wave drag of a Sears-Haack body

Fig 5



Fig 5 Variation in accuracy with N_y and N_z



Fig 6 'Single bulge' body

Fig 6

Fig 7



Fig 7 Variation of wave drag with Mach number for the body of Fig 6



Fig 8 Wave drag error for single bulge body of Ref 8

Fig 9



Fig 9 Wave drag estimates of 10° and 15° cone-cylinders with base area as reference area (after Lighthill[10])



Fig 10 Drag coefficient of a closed axisymmetric body

Fig 11



Fig 11 Wave drag of a smooth axisymmetric body truncated at $x = x_L$



Fig 12 Percentage error in the SAR wave drag when compared with the method of characteristics



Fig 13 Wave drag of a smooth axisymmetric body for a range of thickness


FILMED
8